
Key to Final Exam S3
Computer Architecture

Duration: 1 hr 30 min
Write answers only on the answer sheet.

Exercise     1  (4 points)
Complete the table shown on the  answer sheet. Write down the new values of the registers (except the
PC) and memory that are modified by the instructions. Use the hexadecimal representation. Memory
and registers are reset to their initial values for each instruction.

Initial values: D0 = $FFFF0020  A0 = $00005000  PC = $00006000

D1 = $00000004  A1 = $00005008
D2 = $FFFFFFF0  A2 = $00005010

$005000  54 AF 18 B9 E7 21 48 C0
$005008  C9 10 11 C8 D4 36 1F 88
$005010  13 79 01 80 42 1A 2D 49

Exercise     2  (3 points)
Complete the table shown on the answer sheet. Give the result of the additions and the values of the N, Z,
V and C flags.

Exercise     3  (4 points)
Let us consider the following program. Complete the table shown on the answer sheet.

Main        move.l  #$9507,d7

next1       moveq.l #1,d1
            tst.l   d7
            bpl     next2
            moveq.l #2,d1

next2       moveq.l #1,d2
            cmp.b   #$80,d7
            ble     next3
            moveq.l #2,d2

next3       clr.l   d3
            move.w  #$255,d0
loop3       addq.l  #1,d3
            subq.b  #1,d0
            bne     loop3

next4       clr.l   d4
            move.l  #$93524,d0
loop4       addq.l  #1,d4
            dbra    d0,loop4      ; DBRA = DBF

quit        illegal

Exercise     4  (9 points)
All questions in this exercise are independent. Except for the output registers, none of the data or ad-
dress registers must be modified when the subroutine returns. A string of characters always ends with
a null character (the value zero). For the whole exercise, we assume that the strings of characters are
never empty (they contain at least one character different from the null character).

1. Write down the IsNumber subroutine that determines whether a string contains only digits.
Input: A0.L points to a string that is not empty.
Output: If the string contains only digits, D0.L returns 0.

Otherwise, D0.L returns 1.

2. Write down the GetSum subroutine that adds up all the digits contained in a string of characters.
Input: A0.L points to a string that is not empty and that contains only digits.
Output: D0.L returns the sum of the digits.

Example :

A0 → '7' '0' '4' '8' '9' '4' '2' '0' '3' 0

D0 should return 37 (37 = 7 + 0 + 4 + 8 + 9 + 4 + 2 + 0 + 3).

Tips :
Use a loop that for each character of the string:

➔ Copies the current character in D1.B.
➔ Converts the character into an integer.
➔ Adds the integer to D0.L.

3. By using the IsNumber and GetSum subroutines, write down the CheckSum subroutine that returns the
sum of the digits contained in a string of characters.
Input: A0.L points to a string that is not empty.
Output: If the string contains only digits: D0.L returns 0 and D1.L returns the sum.

Otherwise: D0.L returns 1 and D1.L returns 0.

Last name: ............................................. First name: ........................................... Group: ............................ 

ANSWER SHEET TO BE HANDED IN

Exercise     1  

Instruction Memory Register

Example $005000  54 AF 00 40 E7 21 48 C0
A0 = $00005004
A1 = $0000500C

Example $005008  C9 10 11 C8 D4 36 FF 88 No change

MOVE.L $5006,(A1)+ $005008  48 C0 C9 10 D4 36 1F 88 A1 = $0000500C

MOVE.L #63,2(A1) $005008  C9 10 00 00 00 3F 1F 88 No change

MOVE.B 1(A2),-6(A2,D1.L) $005008  C9 10 11 C8 D4 36 79 88 No change

MOVE.W -8(A1),$12(A1,D2.W) $005008  C9 10 54 AF D4 36 1F 88 No change

Exercise     2  

Operation
Size
(bits)

Result
(hexadecimal)

N Z V C

$59 + $A4 8 $FD 1 0 0 0

$7F8C + $24A6 16 $A432 1 0 1 0

$FFFFFFFF + $EEEEEEEE 32 $EEEEEEED 1 0 0 1

Exercise     3  

Values of registers after the execution of the program.
Use the 32-bit hexadecimal representation.

D1 = $00000001 D3 = $00000055

D2 = $00000002 D4 = $00003525

Exercise     4  

IsNumber            move.l  a0,-(a7)

\loop               move.b  (a0)+,d0
                    beq     \number
                    
                    cmpi.b  #'0',d0
                    blo     \notANumber
                    
                    cmpi.b  #'9',d0
                    bls     \loop
                    
\notANumber         moveq.l #1,d0
                    bra     \quit

\number             clr.l   d0
\quit               movea.l (a7)+,a0
                    rts

GetSum              movem.l a0/d1,-(a7)

                    clr.l   d0
                    clr.l   d1
                    
\loop               move.b  (a0)+,d1
                    beq     \quit
                    
                    sub.b   #'0',d1
                    add.l   d1,d0
                    bra     \loop
                    
\quit               movem.l (a7)+,a0/d1
                    rts

CheckSum            jsr     IsNumber
                    tst.l   d0
                    bne     \notANumber
                    
\number             jsr     GetSum
                    move.l  d0,d1
                    clr.l   d0
                    rts
                    
\notANumber         clr.l   d1
                    rts

